```
> # Maple solution to the Parachutist Problem.
  # Recall that the model changes, since the coefficient of
  friction changes
  # when the parachute is open.
```

```
> #Free fall model for before the parachute is open.
  m:=(125+35)/32;  g:=32;  gamma_1:=1/2;
  DE01:=diff(v(t),t)=g-(gamma_1/m)*v(t);
```

$$m := 5$$

$$g := 32$$

$$gamma\_1 := \frac{1}{2}$$

$$DE01 := \frac{\mathrm{d}}{\mathrm{d}t} v(t) = 32 - \frac{1}{10} v(t) \tag{1}$$

```
> #Free fall model for after the parachute is open:
  gamma_2:=10;
  DE02:=diff(v(t),t)=g-(gamma_2/m)*v(t);
```

$$gamma\_2 := 10$$

$$DE02 := \frac{\mathrm{d}}{\mathrm{d}t} v(t) = 32 - 2\, v(t) \tag{2}$$

```
> #Solve the first DE- The parachutist "fell" from the plane, so
  initial velocity is 0.
  V1:=rhs(dsolve({DE01,v(0)=0},v(t)));
```

$$V1 := 320 - 320\, \mathrm{e}^{-\frac{1}{10} t} \tag{3}$$

```
> #For the second part of the solution, the velocities at t=15
  should match:
  v15:=subs(t=15,V1);
  V2:=dsolve({DE02,v(15)=v15},v(t));
```

$$v15 := 320 - 320\, \mathrm{e}^{-\frac{3}{2}}$$

$$V2 := v(t) = 16 - \frac{\mathrm{e}^{-2t}\left(-304 + 320\, \mathrm{e}^{-\frac{3}{2}}\right)}{\mathrm{e}^{-30}} \tag{4}$$

```
> V2a:=simplify(rhs(V2));
```

$$V2a := -320\, \mathrm{e}^{-2t + \frac{57}{2}} + 304\, \mathrm{e}^{-2t + 30} + 16 \tag{5}$$

```
> # We'll write the velocity function as a piecewise defined
  function for plotting
  Vt:=piecewise(t<=15,V1,t>15,V2a);
```

$$Vt := \begin{cases} 320 - 320\, \mathrm{e}^{-\frac{1}{10} t} & t \le 15 \\ -320\, \mathrm{e}^{-2t + \frac{57}{2}} + 304\, \mathrm{e}^{-2t + 30} + 16 & 15 < t \end{cases} \tag{6}$$

```
> plot(Vt,t=0..20);
```




```
> # We can now get her position at time t.  Initially, she was at
  15,000 feet.  Is this correct?
  S1:=int(Vt,t);
```

$$S1 := \begin{cases} 320\,t + 3200\,e^{-\frac{1}{10}t} & t \le 15 \\ 160\,e^{-2t+\frac{57}{2}} - 152\,e^{-2t+30} + 16\,t + 4712 + 3040\,e^{-\frac{3}{2}} & 15 < t \end{cases} \quad (7)$$

```
> plot(S1,t=0..20);
```




```
> # This is not correct- We should start at 15,000 and decrease in
  height.  This model has us increasing our height. Going back to
  the model, this is the correct way to compute it:
```

```
> S1:=int(V1,t)+C;
```

$$S1 := 320\,t + 3200\,e^{-\frac{1}{10}t} + C \tag{8}$$

```
> simplify(solve(subs(t=0,S1)=-15000,C));
```

$$-18200 \tag{9}$$

```
> S1:=int(V1,t)-18200;
```

$$S1 := 320\,t + 3200\,e^{-\frac{1}{10}t} - 18200 \tag{10}$$

```
> S2:=int(V2a,t)+C2;
```

$$S2 := 160\,e^{-2t+\frac{57}{2}} - 152\,e^{-2t+30} + 16\,t + C2 \tag{11}$$

```
> solve( subs(t=15,S1)=subs(t=15,S2),C2);
```

$$3040\,e^{-\frac{3}{2}} + 152\,e^{0} - 13640 \tag{12}$$

```
> C2:=simplify(%);
```

$$C2 := 3040\, e^{-\frac{3}{2}} - 13488 \tag{13}$$

```
> S2:=int(V2a,t)+C2;
```

$$S2 := 160\, e^{-2t+\frac{57}{2}} - 152\, e^{-2t+30} + 16\, t + 3040\, e^{-\frac{3}{2}} - 13488 \tag{14}$$

```
> St:=piecewise(t<=15,S1,t>15,S2);
```

$$St := \begin{cases} 320\, t + 3200\, e^{-\frac{1}{10}t} - 18200 & t \le 15 \\ 160\, e^{-2t+\frac{57}{2}} - 152\, e^{-2t+30} + 16\, t + 3040\, e^{-\frac{3}{2}} - 13488 & 15 < t \end{cases} \tag{15}$$

```
> plot(St,t=0..120);
```




```
> solve(16*t+C2=0,t);
```

$$-190\, e^{-\frac{3}{2}} + 843 \tag{16}$$

```
> evalf(%);
```

$$800.6052696 \tag{17}$$